# $Id: PKGBUILD 136002 2011-08-21 09:17:29Z eric $
# Maintainer: Eric Bélanger <eric@archlinux.org>

pkgname=gnome-alsamixer
pkgver=0.9.6
pkgrel=4
pkgdesc="Gnome ALSA mixer"
arch=('i686' 'x86_64')
url="https://launchpad.net/gnome-alsamixer"
license=('GPL')
depends=('libgnomeui')
source=(ftp://ftp.archlinux.org/other/gnome-alsamixer/${pkgname}-${pkgver}.tar.gz
        gnome-alsamixer-0.9.6-gtk24.patch gnome-alsamixer.desktop)
md5sums=('387623cfeb079d78325eed2d6bc94251'
         '82b95a0df43429da5098b4da883f6bd0'
         'c7dbc44207bede96e4fafc24a7c7beb3')
sha1sums=('faece601a4329d2b01b9efe65200021538fc066d'
          'b4e3b871676025600170ba54f59db4977b853c08'
          '31639f7932d088ca76a18c79839682be811f23d0')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  patch -Np1 -i "${srcdir}/gnome-alsamixer-0.9.6-gtk24.patch"
  ./configure --prefix=/usr --sysconfdir=/etc --localstatedir=/var
  make
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  make DESTDIR="${pkgdir}" install
  install -D -m644 "${srcdir}/gnome-alsamixer.desktop" \
    "${pkgdir}/usr/share/applications/gnome-alsamixer.desktop"
}
